
137

APPENDIX THREE

THE PROGRAMS FOR PARAFAC (ORTH.) AND
TUCKER2 FOR THE SHENANDOAH EXAMPLE

*Program for the PARAFAC with orthogonal variates and the Tucker2;

*First the streamwater variables are standardized;

libname shen 'c:\prg\shen';
%macro first;
data n&L.; set shen.table&L.;
keep site disch cond ph temp ca mg na k alk so4 c1 si no3 nh4;
disch=v3; cond=v4; ph=v5; temp=v6; ca=v7; mg=v8; na=v9; k=v10; alk=v11;
so4=v12; c1=v13; si=v14;
if v15='<1' then no3=0.5; else no3=v15;if v16='<1' then nh4=0.5; else nh4=v16;
if site=2032589 or site=1628910 or site=1630542 then zzz=1;
else output;

proc standard data=n&L. replace mean=0 out=nm&L.;   run;
proc corr data=nm&L. noprint nocorr cov out=st&L.(type=cov); run;
%mend first;
%let L=a; %first %let L=b; %first %let L=c; %first
%let L=d; %first %let L=e; %first %let L=f; %first

proc iml;
use sta; read all var {site disch cond ph temp ca mg na k alk so4 c1 si
 no3 nh4} into stna;
use stb; read all var {site disch cond ph temp ca mg na k alk so4 c1 si
 no3 nh4} into stnb;
use stc; read all var {site disch cond ph temp ca mg na k alk so4 c1 si
 no3 nh4} into stnc;
use std; read all var {site disch cond ph temp ca mg na k alk so4 c1 si
 no3 nh4} into stnd;
use ste; read all var {site disch cond ph temp ca mg na k alk so4 c1 si
 no3 nh4} into stne;
use stf; read all var {site disch cond ph temp ca mg na k alk so4 c1 si
 no3 nh4} into stnf;

*The streamwater variables are standardized such that the total variance
 for each variable over the six occasions is one.;

ssva=(stna+stnb+stnc+stnd+stne+stnf)/6;  ssvar=ssva[2:15,2:15];
store ssvar;   quit iml;

%macro mann;

proc sort data=n&L.; by site; proc sort data=shen.table2; by site; run;
data table2n; set shen.table2;
keep site anti hamp wev cat sr pedl or ab2400 dd ew dev;
anti=v4; hamp=v5; wev=v6; cat=v7; sr=v8; pedl=v9; or=v10; ab2400=v11;
dd=v12; ew=v13; dev=v14;
run;

*The geological variables are merged with the streamwater variables.;

data t2&L.; merge table2n(in=tab2) n&L.(in=&L.); by site; drop site sr wev;
if &L.=1 then output t2&L.;

*Covariance matrices are created for each occasion.;

proc corr data=t2&L. nocorr noprint cov out=mad(type=cov); run;
proc iml; use mad; load;
read all var {anti hamp cat pedl or ab2400 dd ew dev
disch cond ph temp ca mg na k alk so4 c1 si no3 nh4} into cov&M.;
ssvart=inv(root(diag(ssvar)));
sxx&L.=cov&M.[1:9,1:9];
syy&L.=ssvart*cov&M.[10:23,10:23]*ssvart;
sxy&L.=cov&M.[1:9,10:23]*ssvart;

*The matrices c1-c6 are what will be modeled;

c&M.=inv(root(sxx&L.)`)*sxy&L.;  store c&M. cov&M.;
%mend mann;

%let L=a; %let M=1; %mann
%let L=b; %let M=2; %mann
%let L=c; %let M=3; %mann
%let L=d; %let M=4; %mann
%let L=e; %let M=5; %mann
%let L=f; %let M=6; %mann

m=0;    load;

*The PARAFAC model with orthogonality constraints follows.  K is the matrix
 of orthogonal variates corresponding to the geological variables.  L is the
 matrix of orthogonal variates corresponding to the streamwater variates.
 The K and L that immediately follow are matrices of initial values.;

K={1 0 0 0, 0 1 0 0, 0 0 1 0,
   0 0 0 1, 0 0 0 0, 0 0 0 0,
   0 0 0 0, 0 0 0 0, 0 0 0 0};

L={1 0 0 0, 0 1 0 0, 0 0 1 0,
   0 0 0 1, 0 0 0 0, 0 0 0 0,
   0 0 0 0, 0 0 0 0, 0 0 0 0,
   0 0 0 0, 0 0 0 0, 0 0 0 0,
   0 0 0 0, 0 0 0 0};

*Below follows the alternating least squares algorithm.  K and L are
 at each iteration estimated in a regression that assumes that the rest of
 the parameters are fixed.;

%let m=1; %let n=6;
%macro diag; %do i=&m %to &n; D&i=(diag(K`*C&i*L));
%end; %mend; %diag

ssb=1000000000;

do until (abscrit<0.000001);

%macro sumU;
U=0;
%do i=&m %to &n;
U= U+C&i*L*D&i; %end; %mend;  %sumu

%macro sumV;
V=0;
%do i=&m %to &n;
V= V+C&i`*K*D&i; %end; %mend;  %sumv

K=U*inv(root(U`*U));
L=V*inv(root(V`*V));
          ssa=0;
%macro diagssa; %do i=&m %to &n; D&i=(diag(K`*C&i*L));
ssa=ssa + trace(C&i`*C&i) - 2#trace(C&i`*K*D&i*L`) + trace(D&i**2);
%end; %mend; %diagssa

m=m+1;  crit=ssb-ssa;  ssb=ssa;
        abscrit=abs(crit);
print m ssa crit abscrit;
end;

print K L;

*This step just prints the core matrix;

e=j(4,1,1);
D1=d1*e; D2=d2*e; D3=d3*e; D4=d4*e; D5=d5*e; D6=d6*e;
print D1 D2 D3 D4 D5 D6;

*Next follows the Tucker2 model with four geological variable and four
 streamwater variable components.  G is the matrix of orthogonal variates
 corresponding to the geological variables.  H is the matrix of orthogonal
 variates corresponding to the streamwater variates.;

p=0; ssb=0;

G={1 0 0 0, 0 1 0 0, 0 0 1 0, 0 0 0 1, 0 0 0 0, 0 0 0 0, 0 0 0 0, 0 0 0 0,
   0 0 0 0};
H={1 0 0 0, 0 1 0 0, 0 0 1 0, 0 0 0 1, 0 0 0 0, 0 0 0 0, 0 0 0 0, 0 0 0 0,
   0 0 0 0, 0 0 0 0, 0 0 0 0, 0 0 0 0, 0 0 0 0, 0 0 0 0};

%let m=1; %let n=6;  Gold=G; Hold=H; m=&m; n=&n;

do until (abscrit<0.00000001);

%macro sumU;
U=0;
%do i=&m %to &n;
U= U+C&i*H*H`*C&i`; %end; %mend;  %sumu
G=U*G*inv(root(G`*(U**2)*G));

%macro sumV;
V=0;
%do i=&m %to &n;
V= V+C&i`*G*G`*C&i; %end; %mend;  %sumv
H=V*H*inv(root(H`*(V**2)*H));

%macro sumsqu;
ssa=0;
%do i=&m %to &n;     D&i=G`*C&i*H;  Ce&i=G*D&i*H`;
ssa=ssa + trace((C&i-Ce&i)`*(C&i-Ce&i));
%end; %mend; %sumsqu

Gcrit=ssq(G-Gold); Hcrit=ssq(H-Hold);
Gold=G; Hold=H;
p=p+1;  crit=ssb-ssa;  ssb=ssa;
        abscrit=abs(crit);

print p ssa crit Gcrit Hcrit; end;
print G H D1 D2 D3 D4 D5 D6;


